

## Investigation of Fokker - Planck - Kolmogorov equations for Markov processes in the Maple package

<sup>1</sup>Gurban R. Gasimov, <sup>2</sup>Etibar A. Rzayev, <sup>3</sup>Mirvari H. Aghayeva

<sup>1,2,3</sup>Assoc. Prof., Baku State University, Baku, Azerbaijan..

Corresponding Author: Mirvari H. Aghayeva

**ABSTRACT:** The work is devoted to the study of the Fokker-Planck-Kolmogorov equations for the conditional probability density of Markov processes (Brownian motion processes, fluctuation processes in radio engineering, physical kinetics processes, etc.) in the environment of the Maple mathematical package. Analytical solutions of these equations with initial conditions with a delta function or with a known distribution density are obtained by methods of characteristic functions and integral transformations for various practically important forms of equation coefficients that determine the features of the processes under consideration. Using the constructed 2D, 3D and animated graphs and using the theory of random processes and theory of normal, gamma and beta distributions, the probabilistic distributions of ordinates of the studied processes are determined.

**AMS SUBJECT CLASSIFICATION :** 60J25 , 60E05 , 35K20

**KEY WORDS:** Markov processes, Fokker-Planck-Kolmogorov equations, Random processes, Maple mathematical package

Date of Submission: 01-12-2023

Date of acceptance: 12-12-2023

### I. INTRODUCTION AND PROBLEM STATEMENT

The complete probabilistic characteristic of the Markov process  $U(t)$  is obtained if the function  $F(t, x : \tau, y)$  is known, which represents the probability that at moment  $\tau$  the random variable  $U(t)$  will take a value less than  $y$ , provided that at moment  $t(t < \tau)$   $U(\tau) = x$ . It is known that if there are partial

derivatives  $\frac{\partial}{\partial x}(F), \frac{\partial^2}{\partial x^2}(F)$  and limits

$$\lim_{\Delta t \rightarrow 0} \frac{1}{\Delta t} \int_{|y-x| \geq \delta} (y-x) dF(t-\Delta t, x; t, y) = a(t, x),$$

$$\lim_{\Delta t \rightarrow 0} \frac{1}{\Delta t} \int_{|y-x| \geq \delta} (y-x)^2 dF(t-\Delta t, x; t, y) = b(t, x), \forall \delta > 0,$$

Then  $F(t, x : \tau, y)$  satisfies the equation:

$$\frac{\partial}{\partial t}(F) = -a(t, x) \frac{\partial}{\partial x}(F) - \frac{b(t, x)}{2} \frac{\partial^2}{\partial x^2}(F).$$

If there is a distribution density  $f(t, x : \tau, y) = \frac{\partial}{\partial y}(F)$  and the functions

$\frac{\partial}{\partial \tau}(f), \frac{\partial}{\partial y}(a \cdot f), \frac{\partial^2}{\partial y^2}(b \cdot f)$  are continuous, then  $f$  satisfies equations:

$$\frac{\partial}{\partial t}(f(t, x)) + a(t, x) \frac{\partial}{\partial x}(f(t, x)) + \frac{1}{2} b(t, x) \frac{\partial^2}{\partial x^2}(f(t, x)) = 0 \tag{1}$$

( $f$  – as a function of the initial state parameters)

$$\frac{\partial}{\partial \tau}(f(\tau, y)) + \frac{\partial}{\partial y}(a(\tau, y)f(\tau, y)) - \frac{1}{2} \frac{\partial^2}{\partial y^2}(b(\tau, y)f(\tau, y)) = 0 \tag{2}$$

( $f$  – as a function of the final state parameters)

Equations (1) and (2) are known as the "Fokker–Planck- Kolmogorov" (FPK); FPK equation represents the evolution in time and space of the probability density function of the states of a stochastic dynamical system. The purpose of this article is to determine the analytical solutions of these equations in practically important cases of coefficients  $a(t, x)$  and  $b(t, x)$  and the probabilistic interpretation of the solutions obtained in the environment of the mathematical package Maple.

These equations belong to the type of parabolic equations. In order for an equation of this type to uniquely determine its solution, it is necessary to have initial and boundary conditions for the desired function. When  $\tau = t$ , the difference between the random variable  $X = U(t)$  and  $Y = U(\tau)$  disappears, and, therefore, the conditional density  $f(t, x : \tau, y)$  should be zero for any  $y \neq x$ . On the other hand, the normalization

condition  $\int_{-\infty}^{\infty} f(t, x : \tau, y) dy = 1$  should remain valid for  $t = \tau$ , and this may be case if  $f$  turns into a delta-function, i.e. the condition must be met:  $f(t, x : \tau, y)|_{\tau=t} = \delta(x - y)$ .

The boundary conditions are determined by the requirement that the probability density vanishes at the boundaries of the region of variation of a random process; if the interval of variation  $y$  is equal to  $(-\infty, \infty)$ , then for any  $\tau$ ,  $f$  must vanish at  $|y| \rightarrow \infty$ . In the case when the range of  $(y_1, y_2)$  - possible values of the ordinates of a random process is limited, the conditions must be met:

$$\left[ B_y f \equiv a(\tau, y)f(\tau, y) - \frac{1}{2} \frac{\partial}{\partial y}(b(\tau, y)f(\tau, y)) \right]_{y=y_i} = 0, i = 1, 2. \tag{3}$$

CASE 1:  $a(\tau, y) = y, \quad b(\tau, y) = y^2 \quad (0 < y < \infty)$

In this case the initial ordinate of the process is given as a random function with a known density:

$$f_0(y) = \begin{cases} 1, & 0 < y < 1, \\ 0, & \text{otherwise} \end{cases}$$

> **PDE1:=diff(f1(tau,y),tau)+diff(y\*f1(tau,y),y)-1/2\*diff(y^2\*f1(tau,y),y\$2)=0;**

$$PDE1 := \frac{\partial}{\partial \tau} f1(\tau, y) - y \left( \frac{\partial}{\partial y} f1(\tau, y) \right) - \frac{y^2 \left( \frac{\partial^2}{\partial y^2} f1(\tau, y) \right)}{2} = 0 \tag{4}$$

> **ic1:=f1(0,y)=piecewise(y>0 and y<1,1,0);**

$$ic1 := f1(0, y) = \begin{cases} 1, & 0 < y < 1, \\ 0, & \text{otherwise} \end{cases} \tag{5}$$

The satisfaction of condition (5) is checked directly :

> **bc1:=limit(y\*f1(tau, y)-(1/2)\*diff(y^2\*f1(tau, y),y),y=0)=0;**  
**bc1 := 0 = 0** (6)

Solving the problem (4)-(5):

> **pdsolve ([PDE1,ic1]);**

$$f1(\tau, y) = m^{-1} \left( \frac{e^{\frac{s(s-1)\tau}{2}}}{s}, s, y \right) \tag{7}$$

The solution is obtained in the form of the inverse Mellin transform. To go to the original, we use some properties of this transformation and bring the expression under  $(m)^{-1}$  to a form suitable for the application of these properties :

$$m(f(y), y, s) = \int_0^{\infty} y^{s-1} f(y) dy \equiv \tilde{f}(s), s > 0$$

$$\begin{aligned} m^{-1} \left( \frac{\tilde{f}(s+1)}{s} \right) &= \int_y^{\infty} f(y1) d(y1), m^{-1}(\tilde{f}1(s)\tilde{f}2(s)) = \int_0^{\infty} \frac{f1\left(\frac{y}{y1}\right) f2(y1) d(y1)}{y1} \\ \exp\left\{\frac{(s^2 - s)\tau}{2}\right\}/s &= \exp\left\{-\frac{(s+1)\tau}{2}\right\}/s * \exp\left\{\frac{(s)^2 \tau}{2}\right\} \exp\left\{\frac{\tau}{2}\right\} \equiv \\ &\equiv \left(\tilde{f}1(s+1)/s\right)\tilde{f}2(s) \exp\left(\frac{\tau}{2}\right) \end{aligned}$$

> **addtable(mellin,f(y),F(s),y,s);**  
 > **assume(s>0,tau>0);**  
 > **A0:=unapply(invmellin(exp(-s\*tau/2),s,y),tau,y);**

$$A0 := (\tau \sim, y) \mapsto \frac{2^{-1} Dirac\left(y^{\frac{2}{\tau \sim}} - e^{-1}\right)}{\tau \sim} \tag{8}$$

> **A1:=unapply(exp(tau/2)\*int(A0(tau,y1),y1=y..infinity),tau,y);**

$$A1 := (\tau \sim, y) \mapsto e^{\frac{\tau}{2}} \left( \int_y^{\infty} \frac{2^{-1} e^{-1} Dirac\left(y1^{\frac{2}{\tau \sim}} - e^{-1}\right)}{\tau \sim} dy1 \right) \tag{9}$$

> **A2:=unapply(invmellin(exp((s^2\*tau/2)),s,y),tau,y)**

$$A2 := (\tau \sim, y) \mapsto \frac{\sqrt{2} e^{-\frac{\ln\left(\frac{1}{y}\right)^2}{2\tau \sim}}}{2\sqrt{\tau \sim} n} \tag{10}$$

> **assume(y>0);**  
 > **f1:=unapply(simplify(int(A1(tau,y/u)\*A2(tau,u)/u,u=0..infinity)),tau,y);**

$$f1 := (\tau \sim, y \sim) \mapsto \int_0^{\infty} \frac{\sqrt{2} e^{\frac{(-\tau \sim + \ln(u)) \cdot (\tau \sim + \ln(u))}{2\tau \sim}} \left( \int_{\frac{y \sim}{u}}^{\infty} \frac{2e^{-1} Dirac\left(y1^{\frac{2}{\tau \sim}} - e^{-1}\right)}{\tau \sim} dy1 \right)}{2\sqrt{\pi} \sqrt{\tau \sim} u} du \tag{11}$$

An integral representation for the density is obtained; it shows the exponential nature of the distribution with a parameter depending on  $\tau$ . We normalize this density and construct a joint graph for the values  $\tau = 1, \tau = 2$  and  $\exp(-y)$  [ Fig.1].

```
> f1n:=unapply(int(f1(tau,y),y=0..infinity),tau):
> f1nr:=unapply(f1(tau,y)/f1n(tau),tau,y):
> f1s1:=unapply(subs(tau=1,f1nr(tau,y)),y):
> f1s2:=unapply(subs(tau=2,f1nr(tau,y)),y):
> plot([f1s1(y),f1s2(y),exp(-y)],y=0..10,thickness=3,color=[red,blue,green]);
```




Fig.1

```
> evalf(int(f1s1(y),y=0..infinity));
1.0 (12)
```

CASE 2:  $a(\tau, y) = \alpha_0 + \alpha_1 * y, b(\tau, y) = \beta_0$

In this case we solve a problem with condition of the form of a,  $\delta(x, y)$  we use method of characteristic

functions; denote  $\psi(\tau, z) = \ln \left( \int_{-\infty}^{\infty} e^{izy} f(\tau, y) dy \right)$ .

With respect to  $\psi$  (psi2), the following problem is obtained :

```
> PDE2:=diff(psi2(tau,z),tau)+(beta0/2)*z^2+I*alpha0*z-alpha1*z*diff(psi2(tau,z),z)=0;
PDE2 := \frac{\partial}{\partial \tau} \psi_2(\tau, z) + \frac{\beta_0 z^2}{2} + I \alpha_0 z - \alpha_1 z \left( \frac{\partial}{\partial z} \psi_2(\tau, z) \right) = 0 (13)
```

```
> fourier(Dirac(x-y),y,z);
e^{-Ixz} (14)
```

```
> ic2:=psi2(0,z)=I*x*z;
ic2 := \psi_2(0, z) = Ixz (15)
```

```
> g2:=unapply(pdsolve([PDE2,ic2],tau,z,x,alpha0,alpha1,beta0);
g2 := (\tau, z, x, \alpha_0, \alpha_1, \beta_0) \mapsto \psi_2(\tau, z) =
= \frac{-z^2 \cdot \beta_0 \cdot e^{2\alpha_1 \tau} + 4I(\alpha_1 x - \alpha_0)z e^{\alpha_1 \tau} + \beta_0 z^2 + 4I\alpha_0 z}{4\alpha_1} (16)
```

The last expression is a characteristic function of a normal distribution with parameters

$$\sigma^2 = \frac{\beta_0(e^{2\alpha_1 \tau} - 1)}{2\alpha_1}, \quad \mu = \frac{((\alpha_1 x - \alpha_0)e^{\alpha_1 \tau} + \alpha_0)}{\alpha_1}$$

```
> pdetest(g2(tau,z,x,alpha0,alpha1,beta0),[PDE2,ic2]);
[0,0] (17)
```

Consider a special sub-case:  $x=1, \alpha_0 = \alpha_1 = \beta_0 = 1$  and by the inverse Fourier transform we determine the density of the distribution:

> `g2s:=unapply(subs({x=1,alpha0=1,alpha1=1,beta0=1},g2(tau,z,x,alpha0,alpha1,beta0)),tau,z);`

$$g2s := (\tau \sim, z) \mapsto \psi2(\tau \sim, z) = -\frac{z^2 e^{2\tau \sim}}{4} + \frac{z^2}{4} + Iz \quad (18)$$

> `assume(tau>0);`

> `f2s:=unapply(invfourier(exp(-z^2*(exp(2*tau)-1)/4+I*z),z,y),tau, y);`

$$f2s := (\tau \sim, y \sim) \mapsto \frac{e^{\frac{(y \sim + 1)^2}{e^{2\tau \sim} - 1}} \sqrt{\frac{\pi}{e^{2\tau \sim} - 1}}}{\pi} \quad (19)$$


We check the normalization condition for the resulting density :

> `simplify(int(f2s(tau,y),y=-infinity..infinity));`

$$1 \quad (20)$$

Let is construct a part of the surface of the function  $f2s(\tau, y)$  in the space  $(\tau, y)$  [Fig.2]:


> `plot3d(f2s(tau,y),tau=0.1..0.5,y=-5..5,color=sin(tau));`



**Fig.2**

We select individual sections of the resulting surface [Fig.3]:

> `plot([subs(tau=0.1,f2s(tau,y)),subs(tau=0.5,f2s(tau,y))],y=-5..5,thickness=5,color=[red,blue]);`



**Fig.3**

> `int(subs(tau=0.1,f2s(tau,y)),y=-infinity..infinity);`

$$1. \quad (21)$$

> `int(subs(tau=0.5,f2s(tau,y)),y=-infinity..infinity);`

$$1. \quad (22)$$

CASE 3:  $a(\tau, y) = \alpha0$ ,  $b(\tau, y) = \beta0 + \beta1 * y$

This case is also investigated with a delta-initial condition and using the method of characteristic functions; a similar problem is obtained for the function  $\psi(psi3)$  :

> `PDE3:=diff(psi3(tau,z),tau)+I/2*beta1*z^2*diff(psi3(tau,z),z)=-I*alpha0*z-1/2*beta0*z^2;`

$$PDE3 := \frac{\partial}{\partial \tau} \psi3(\tau, z) + \frac{I\beta1z^2 \left( \frac{\partial}{\partial z} \psi3(\tau, z) \right)}{2} = -I\alpha0z - \frac{\beta0z^2}{2} \quad (23)$$

> **ic3:=psi3(0,z)=-I\*x\*z;**

$$ic3 := \psi3(0, z) = -Ixz \tag{24}$$

> **dsolve([PDE3,ic3]);**

$$\begin{aligned} \psi3(\tau, z) = & \frac{I\beta0z}{\beta1} - \frac{1}{\beta1} \left( I \left( I \ln \left( \frac{4Iz\beta1\tau + 4}{z^2} \right) \alpha0 - I \ln \left( \frac{(Iz\beta1\tau + 2)^2}{z^2} \right) \alpha0 + \right. \right. \\ & \left. \left. + \frac{2\beta1xz}{Iz\beta1\tau + 2} + \frac{2\beta0z}{Iz\beta1\tau + 2} + 2 \arctan \left( \frac{z\beta1\tau}{Iz\beta1\tau + 2} \right) \alpha0 \right) \right) \end{aligned} \tag{25}$$

For the convenience of probabilistic analysis, in the last expression, we express the arc tan in terms of the ln:

>**A:=convert(arctan(beta1\*tau\*z/(I\*beta1\*tau\*z+2)),ln);**

$$A := \frac{I}{2} \left( \ln \left( \frac{2}{Iz\beta1\tau + 2} \right) - \ln \left( 1 + \frac{I\beta1\tau z}{Iz\beta1\tau + 2} \right) \right) \tag{26}$$

>**g3:=unapply(I\*beta0\*z/beta1+alpha0\*ln((4+4\*I\*beta1\*tau\*z)/(I\*beta1\*tau\*z+2)^2)/beta1-I\*2\*x\*z/(I\*beta1\*tau\*z+2)-2\*I\*beta0\*z/(beta1\*(I\*beta1\*tau\*z+2))-I\*2\*alpha0\*A/beta1,tau,z,alpha0,beta0,beta1);**

$$\begin{aligned} g3 := (\tau, z, \alpha0, \beta0, \beta1) \mapsto & \frac{I\beta0z}{\beta1} + \frac{\alpha0 \ln \left( \frac{4Iz\beta1\tau + 4}{(Iz\beta1\tau + 2)^2} \right)}{\beta1} - \frac{2Ixz}{Iz\beta1\tau + 2} \\ & - \frac{2I\beta0z}{(Iz\beta1\tau + 2)\beta1} + \frac{\alpha0 \ln \left( \ln \left( \frac{2}{Iz\beta1\tau + 2} \right) - \ln \left( 1 + \frac{I\beta1\tau z}{Iz\beta1\tau + 2} \right) \right)}{\beta1} \end{aligned} \tag{27}$$

Consider a special sub-case  $x = -1, \alpha0 = \beta0 = \beta1 = 1$  and by the inverse Fourier transform, we determine the density:

>**exp(subs({alpha0=1,beta0=1,beta1=1,x=-1},g3(tau,z,alpha0,beta0,beta1)));**

$$e^{Iz + \ln \left( \frac{4Iz\tau + 4}{(Iz\tau + 2)^2} \right) + \ln \left( \frac{2}{Iz\tau + 2} \right) - \ln \left( 1 + \frac{I\tau}{Iz\tau + 2} \right)} \tag{28}$$

> **f3:=unapply (invfourier(%z, y),tau, y);**

$$\begin{aligned} f3 := (\tau, y) \mapsto & \frac{2e^{-\frac{2(y+1)}{\tau}} (y+1) \left( 2Heaviside(y+1) - 1 + \text{signum} \left( 0, R \left( \frac{1}{\tau} \right), 0 \right) \right)}{\tau^2} \end{aligned} \tag{29}$$

We calculate mathematical expectation and the variance of the distribution with density  $f3(\tau, y)$

> **assume(tau>0);**

> **ef3:=unapply(simplify(int(y\*f3(tau,y),y=-infinity..infinity)),tau);**

$$ef3 := \tau \rightsquigarrow \tau \sim -1 \tag{30}$$

>**df3:=unapply(simplify(int(y^2\*f3(tau,y),y=-infinity..infinity)) - ef3(tau)^2,tau);**

$$df3 := \tau \rightsquigarrow \frac{3\tau \sim^2}{2} - \tau \sim + 1 - (\tau \sim - 1)^2 \tag{31}$$


> **f3s1:=unapply(subs(tau=1,f3(tau,y)),y);**

$$f3s1 := y \mapsto 4e^{-2y-2} (y+1) Heaviside(y+1) \tag{32}$$

```
> f3s2:=unapply(subs(tau=2,f3(tau,y)),y);
    f3s2 := y ↦ e-y-1(y+1)Heaviside(y+1) (33)
```

Let's construct graphs of special case and  $f_3(\tau, y)$  at  $\tau = 1$  and  $\tau = 2$  [Fig.4].

```
> plot([f3s1(y),f3s2(y)],y=-5..5,thickness=3,color=[red,blue]);
```




**Fig.4**

The resulting graphs are of shifted gamma-distribution densities at different values of the distribution parameters. We will check the normalization condition for  $f_{3s1}$  and  $f_{3s2}$  and construct a part of the 3D-surface  $f_3(\tau, y)$  [Fig.5]:

```
> int(f3s1(y),y=-infinity..infinity);
    1 (34)
```

```
> int(f3s2(y),y=-infinity..infinity);
    1 (35)
```

```
> plot3d(f3(tau,y),tau=1..2,y=-5..5,color=sin(tau));
```



**Fig.5**

We introduce a gamma-distributed random variable and find expressions of its numerical characteristics through the numerical characteristics of the  $f_3$ -distribution [Fig .6]:

```
> X0:=RandomVariable(GammaDistribution(a,b));
> f0:=unapply(PDF(X0,u),u);
```

$$f_0 := u \mapsto \begin{cases} 0 & u < 0 \\ \frac{\left(\frac{u}{a}\right)^{b-1} e^{-\frac{u}{a}}}{a\Gamma(b)} & \text{otherwise} \end{cases} \quad (36)$$

> Mean(X0);

$$ab \quad (37)$$

> Variance(X0);

$$a^2b \quad (38)$$

$$\begin{cases} \text{mean}(X_0) = ef_3 + 1 \\ \text{variance}(X_0) = df_3 \end{cases} \Rightarrow a = \frac{\tau}{2}, b = 2$$

```
> X1:=RandomVariable(GammaDistribution(1,2));
> X2:=RandomVariable(GammaDistribution(1/2,2));
> plot([PDF(X1,u+1),f3s2(u),PDF(X2,u+1),f3s1(u)],u=-5..5,thickness=
3,style=[point,line,point,line],color=[black,"SkyBlue",red,green] );
```




Fig.6

## II. KUMMER FUNCTIONS IN FPK – EQUATION

Consider the equation for the conditional density with coefficients of case 2, in which the variables  $t$  and  $x$  are parameters

```
> assume(t>=0,tau>t);
> PDE4:=diff(f4(t,x,tau,y),tau)+diff((a0+a1*y)*f4(t,x,tau,y),y)-b0/2*diff(f4(t,x,tau,y),y$2)=0;
```

$$PDE4 := \frac{\partial}{\partial \tau} f_4(t \sim, x, \tau \sim, y \sim) + a_1 f_4(t \sim, x, \tau \sim, y \sim) + (a_1 y \sim + a_0) \times \left( \frac{\partial}{\partial y \sim} f_4(t \sim, x, \tau \sim, y \sim) \right) - \frac{b_0 \left( \frac{\partial^2}{\partial y \sim^2} f_4(t \sim, x, \tau \sim, y \sim) \right)}{2} = 0 \quad (39)$$

We find the general solution of the last equation:

```
> struc4:=pdsolve(PDE4,f4(t,x,tau,y));
```



$$\begin{aligned}
 \text{struc4} &:= f4(t \sim, x, \tau \sim, y \sim) = \_F3(\tau \sim) \_F4(y \sim) \_F6(t \sim, x) \text{ where} \\
 &\left[ \left\{ \frac{-F6(t \sim, x) - c_4}{2} = 0, \frac{d}{d\tau \sim} \_F3(\tau \sim) = \_c_3 \_F3(\tau \sim), \right. \right. \\
 &\left. \left. \frac{d^2}{dy \sim^2} \_F4(y \sim) = \frac{2 \_F4(y \sim) - c_3}{b0} - \frac{c_4}{b0} + \right. \right. \\
 &\left. \left. \frac{2 \left( \frac{d}{dy \sim} \_F4(y \sim) \right) a1 y \sim + 2a1 \_F4(y \sim) + 2 \left( \frac{d}{dy \sim} \_F4(y \sim) \right) a0 \right\}}{b0} \right]
 \end{aligned} \tag{40}$$

> **fk2:=unapply(PDEtools[build](struc4),t,x,tau,y);**

$$\text{fk2} := (t \sim, x, \tau \sim, y \sim) \mapsto f4(t \sim, x, \tau \sim, y \sim) = \_C1e^{-c_3\tau \sim} \_F6(t \sim, x) \_C2 \tag{41}$$

$$\begin{aligned}
 &KummerM\left(\frac{-c_3 + a1}{2a1}, \frac{1}{2}, \frac{(a1y \sim + a0)^2}{a1b0}\right) + \_C1e^{-c_3\tau \sim} \_F6(t \sim, x) \_C3 \\
 &KummerU\left(\frac{-c_3 + a1}{2a1}, \frac{1}{2}, \frac{(a1y \sim + a0)^2}{a1b0}\right)
 \end{aligned}$$

The resulting expression shows that the general solution is represented as a linear combination of the Kummer function with coefficients depending on the parameters t, x and  $\tau$ .

> **Kum:=z\*diff(y(z),z\$2)+(mu-z)\*diff(y(z),z)-nu\*y(z)=0;**

$$\text{Kum} := z \left( \frac{d^2}{dz^2} y \sim(z) \right) + (\mu - z) \left( \frac{d}{dz} y \sim(z) \right) - \nu y \sim(z) = 0 \tag{42}$$

> **dsolve(Kum,y(z));**

$$y \sim(z) = \_C1KummerM(\nu, \mu, z) + \_C2KummerU(\nu, \mu, z) \tag{43}$$

We note that the characteristic function of the beta distribution can be represented through the KummerM function:

> **KR:=RandomVariable(BetaDistribution(p,q));**

$$\text{KR} := \_R2 \tag{44}$$

> **sf:=unapply(PDF(KR,u),u,p,q);**

$$\text{sf} := (u, p, q) \mapsto \begin{cases} 0 & u < 0 \\ \frac{u^{p-1}(1-u)^{q-1}}{B(p, q)} & 0 < u < 1 \\ 0 & \text{otherwise} \end{cases} \tag{45}$$

> **CharacteristicFunction(KR,t);**

$$\text{hypergeom}([p],[p+q],It \sim) \tag{46}$$

> **convert(hypergeom([p],[p+q],I\*t),Kummer);**

$$\text{KummerM}(p, p + q, It \sim) \tag{47}$$

### III. APPLICATION OF THE VARIABLE CONVERSION METHOD

The method of variable transformation is very effective for solving the FPK - equation. If it is possible to convert the FPK- equation to the usual form

$$\frac{\partial}{\partial \tau'} f'(\tau', y') \pm \frac{1}{2} \frac{\partial^2}{\partial y'^2} f'(\tau', y') = 0,$$

then the new independent variable obeys the normal distribution law. The presence of a normal distribution law for  $Y'$  does not mean that the normal distribution law for the ordinates of the random process under study is valid;

normality is preserved only with a linear transformation. We have applied this method to the following important tasks. With the help of conversion  $\tau' = \phi(\tau)$ ,  $y' = \psi(\tau, y)$ , where  $\phi$  and  $\psi$  are arbitrary functions satisfying the requirements of an unambiguous transition from one system of variables to another, reduces to the equation

$$\frac{\partial}{\partial \tau'} f'(\tau', y') + \frac{\partial}{\partial y'} [a'(\tau', y') f'] - \frac{1}{2} \frac{\partial^2}{\partial y'^2} [b'(\tau', y') f'] = 0,$$

where  $a'(\tau', y') = \left[ \frac{1}{2} \frac{\partial^2 \psi(\tau, y)}{\partial y^2} b(\tau, y) + \frac{\partial \psi(\tau, y)}{\partial y} a(\tau, y) + \frac{\partial \psi(\tau, y)}{\partial \tau} \right] : \frac{d\phi(\tau)}{d\tau}$ ,

$$b'(\tau', y') = \left\{ \left[ \frac{\partial \psi(\tau, y)}{\partial y} \right]^2 b(\tau, y) \right\} : \frac{d\phi(\tau)}{d\tau}$$

In the case then  $a(\tau, y) = a(\tau)y + \beta(\tau)$ ,  $b(\tau, y) = \gamma(\tau)$  transformation

$$\psi(\tau, y) = y \exp \left( - \int_0^\tau \alpha(\tau_1) d\tau_1 \right) - \int_0^\tau \beta(\tau_2) \exp \left\{ - \int_0^{\tau_2} \alpha(\tau_1) d\tau_1 \right\} d\tau_2,$$

$$\phi(\tau) = \int_0^\tau \gamma(\tau_2) \exp \left\{ - 2 \int_0^{\tau_2} \alpha(\tau_1) d\tau_1 \right\} d\tau_2 \tag{48}$$

leads to the goal.

Now let us consider the application of formula (48) in two problems:

- 1)  $a(\tau, y) \equiv a(\tau), b(\tau, y) \equiv b(\tau)$ ;
- 2) the problem associated with "white noise";
- 3)  $a(\tau, y) = \tau(y - c), b(\tau, y) = 2\tau(y - c)^2 (c < y < \infty)$ .

> **diff(f(tau,y),tau)+a(tau)\*diff(f(tau,y),y)-1/2\*b(tau)\*diff(f(tau,y),y\$2)=0;**

$$\frac{\partial}{\partial \tau} f(\tau, y) + a(\tau) \left( \frac{\partial}{\partial y} f(\tau, y) \right) - \frac{b(\tau) \left( \frac{\partial^2}{\partial y^2} f(\tau, y) \right)}{2} = 0 \tag{49}$$

According to formulas (4), for the selected special case, we perform the following transformations of variables:

$$y' = y - \int_0^\tau b(z) dz, \tau' = \int_0^\tau b(z) dz \Rightarrow a(\tau) = (\tau)^2, b(\tau) = (\tau)^2 \Rightarrow y' = y - \frac{\tau^3}{3},$$

$$\tau' = \frac{\tau^3}{3} \Rightarrow tr : y = y' + \tau', \tau = (3\tau')^{\frac{1}{3}}$$

> **PDE5:=diff(f5(tau,y),tau)+tau^2\*diff(f5(tau,y),y)-tau^2/2\*diff(f5(tau,y),y\$2)=0;**

$$PDE5 := \frac{\partial}{\partial \tau} f5(\tau, y) + \tau^2 \left( \frac{\partial}{\partial y} f5(\tau, y) \right) - \frac{\tau^2 \left( \frac{\partial^2}{\partial y^2} f5(\tau, y) \right)}{2} = 0 \tag{50}$$

> **tr:={y=y1+tau1,tau=(3\*tau1)^(1/3)};**

$$tr := \{\tau \sim= 3^{1/3} t^{1/3}, y \sim= y1 + \tau1\} \tag{51}$$

After the transformation, the equation PDE5 takes the "classical form":

>PDE5CH:=dchange(tr,PDE5,[tau1,y1],simplify) ;

$$PDES5CH := \frac{3^{3/2} \tau1^{2/3} \left( 2 \frac{\partial}{\partial \tau1} f5(\tau1, y1) - \frac{\partial^2}{\partial y1^2} f5(\tau1, y1) \right)}{2} = 0 \tag{52}$$

>ic5:=f5(0,y1)=Dirac(x1-y1);

$$ic5 := f5(0, y1) = \text{dirac}(x1 - y1) \tag{53}$$

> assume(tau1>0);

> pdsolve([PDE5CH,ic5]);

$$f5(\tau1 \sim, y1) = \frac{e^{-\frac{(x1-y1)^2}{2\tau1}} \sqrt{2}}{2\sqrt{\tau1} \sim \sqrt{\pi}} \tag{54}$$

The resulting solution is the density of the normal distribution and, since the transformation is linear with respect to y , the density for the ordinates of the random process under study also turns out to be normal :

$$f5(\tau1, y1) \left[ y1 = y - \frac{\tau^3}{3}, \tau1 = \frac{\tau^3}{3} \right] \Rightarrow f(\tau, y) = \frac{\exp \left( -\frac{\left( x1 - y + \frac{\tau^3}{3} \right)^2}{\frac{2\tau^3}{3}} \right)}{\sqrt{\frac{2\pi\tau^3}{3}}} \times \times \frac{d}{dy} \left( y - \frac{\tau^3}{3} \right) \tag{55}$$


$$notf5(\tau1 \sim, y1) \left[ y1 = y \sim - \frac{\tau \sim^3}{3}, \tau1 \sim = \frac{\tau \sim^3}{3} \right]$$

Below we present 2D-graphs of auxiliary and main density at different values of  $\tau$  and when  $x1=0$ , as well as their comparative 3D-surface [Fig.7], [Fig.8], [Fig.9].

> f51:=unapply (exp(-y1^2/(2\*tau1))/sqrt (2\*Pi\*tau1),tau1,y1):

>

> plot([f51(1,y1),f51(2,y1),f51(3,y1)],y1=-20..20,thickness=5,color=[red, blue, green]);



**Fig.7**

```
> f5:=unapply(exp(-(y-tau^3/3)^2/(2*tau^3/3))/sqrt(2*Pi*tau^3/3),tau,y);
```

$$f5 := (\tau \sim, y \sim) \mapsto \frac{e^{-\frac{3\left(y\sim - \frac{\tau\sim^3}{3}\right)^2}{2\tau\sim^3}}}{2\tau\sim \sqrt{\tau\sim \pi}} \quad (56)$$


```
> assume(tau>0);
```

```
> int(f5(tau,y),y=-infinity..infinity);
```

1


(57)

```
> plot([f5(1,y),f5(2,y),f5(3,y)],y=-20..20,thickness=5,color=[red, blue,green]);
```



**Fig.8**

```
> plot3d([f51(tau,y),f5(tau,y)],tau=1..3,y=-20..20,color=[red,"SkyBlue"]);
```



**Fig. 9**

2) We turn to the equation related to the problem of determining the distribution law of the ordinate of a random process  $U(t)$ :

$$U'(t) + \alpha U(t) = m\xi(t), U(0) = x,$$

where  $\xi(t)$  is “white noise” having zero expectation and correlation function  $K_\xi(\tau) = \delta(\tau)$ .

```
> assume(alpha>0,m>0);
```

```
> PDEC:=diff(f(tau,y),tau)-alpha*y*diff(f(tau,y),y)-m^2/2*diff(f(tau,y),y$2)=0;
```

$$\begin{aligned}
 PDEC := & \frac{\partial}{\partial \tau} f(\tau, y) - \alpha y \left( \frac{\partial}{\partial y} f(\tau, y) \right) - \\
 & \frac{m^2 \left( \frac{\partial^2}{\partial y^2} f(\tau, y) \right)}{2} = 0
 \end{aligned}
 \tag{58}$$

According to formulas (48), we perform transformations of variables:

$$y' = y^{\alpha\tau}, \tau' = \frac{m^2}{2\alpha} (e^{2\alpha\tau} - 1) \Rightarrow y = \frac{y'}{\sqrt{1 + \frac{2\alpha}{m^2} \tau'}}, \tau = \frac{1}{2\alpha} \ln \left( 1 + \frac{2\alpha}{m^2} \tau' \right)$$

> `tr:={tau=1/(2*alpha)*ln(1+2*alpha/m^2*tau1),y=y1/sqrt(1+2*alpha/m^2*tau1)};`

$$tr := \left\{ \begin{aligned} \tau &= \frac{\ln \left( 1 + \frac{2\alpha \tau_1}{m^2} \right)}{2\alpha}, \\ y &= \frac{y_1}{\sqrt{1 + \frac{2\alpha \tau_1}{m^2}}} \end{aligned} \right\}
 \tag{59}$$

> `PDECH:=dchange(tr,PDEC,[tau1,y1],simplify);`

$$PDECH := \left( \frac{\partial}{\partial \tau_1} f(\tau_1, y_1, -\alpha m) - \frac{\partial^2}{\partial y_1^2} f(\tau_1, y_1, -\alpha m) \right) \times
 \tag{60}$$

$$\times (2\alpha \tau_1 + m^2) = 0$$

> `PDE6:=expand(PDECH/(2*alpha*tau1+m^2));`

$$PDE6 := \frac{\partial}{\partial \tau_1} f(\tau_1, y_1, -\alpha m) - \frac{\partial^2}{\partial y_1^2} f(\tau_1, y_1, -\alpha m) = 0
 \tag{61}$$

We solve the auxiliary equation, plot the solution with a special choice of parameters **[Fig.10]**:

> `assume(x1,real);assume(tau1>0);`

> `icch:=f(0,y1,alpha,m)=Dirac(y1-x1);`

$$icch := f(0, y_1, \alpha, m) = Dirac(x_1 - y_1)
 \tag{62}$$

> `pdsolve([PDE6,icch]);`

$$f(\tau_1, y_1, \alpha, m) = \frac{e^{-\frac{(x_1 - y_1)^2}{2\tau_1}} \sqrt{2}}{2\sqrt{\tau_1} \sqrt{\pi}}
 \tag{63}$$

> `f0:=unapply(exp(-(y1-x1)^2/(2*tau1))/sqrt(2*Pi*tau1),tau1,y1,x1);`

$$f_0 := (\tau_1, y_1, x_1) \mapsto \frac{e^{-\frac{(y_1 - x_1)^2}{2\tau_1}} \sqrt{2}}{2\sqrt{\tau_1} \sqrt{\pi}}
 \tag{64}$$

> `plot(subs({x1=1,tau1=1},f0(tau1,y1,x1)),y1=-5..5,thickness=5,color="SkyBlue");`




Fig.10

We determine the solution of the original equation:

$$f_0(\tau_1, y_1, x_1) \left[ \tau_1 = \frac{m^2 (e^{2\alpha\tau} - 1)}{2\alpha}, y_1 = ye^{\alpha\tau} \right] \frac{\partial}{\partial y} (\psi(\tau, y)) \Rightarrow f_6$$

> tau1n:=unapply(m^2/(2\*alpha)\*(exp(2\*alpha\*tau)-1),tau);

$$\tau_{1n} := \tau \rightsquigarrow \frac{m^2 (e^{2\alpha\tau} - 1)}{2\alpha} \tag{65}$$

> y1n:=unapply(y\*exp(alpha\*tau),tau,y);

$$y_{1n} := (\tau \sim, y \sim) \mapsto y \sim e^{\alpha\tau\sim} \tag{66}$$

> f6:=unapply(subs({tau1=tau1n(tau),y1=y1n(tau,y)},f0(tau1,y1,x1))\*exp(alpha\*tau),tau,y,x1,alpha,m);

$$f_6 := (\tau \sim, y \sim, x_1 \sim, \alpha \sim, m \sim) \mapsto \frac{e^{-\frac{(-x_1\sim + y\sim e^{\alpha\tau\sim})^2}{m^2(e^{2\alpha\tau\sim} - 1)}} e^{\alpha\tau\sim}}{\sqrt{\frac{\pi m^2 (e^{2\alpha\tau\sim} - 1)}{\alpha \sim}}} \tag{67}$$

The resulting expression also reflects the density of the normal distribution [Fig.11].

> plot(subs({x1=1,tau=1,alpha=1,m=1},f6(tau,y,x1,alpha,m)),y=-5..5,thickness=5,color=blue);




Fig.11

> int(f6(1,y,1,1,1),y=-infinity..infinity);

$$1 \tag{68}$$

$$3) a(\tau, y) = \tau(y - c), b(\tau, y) = 2\tau(y - c)^2, y > c$$

> PDE7:=diff(f(tau,y),tau)+tau\*(y-c)\*diff(f(tau,y),y)+tau\*(y-c)^2\* diff(f(tau,y),y\$2)=0;

$$PDE7 := \frac{\partial}{\partial \tau} f(\tau, y) + \tau (y - c) \left( \frac{\partial}{\partial y} f(\tau, y) \right) + \tau (y - c)^2 \left( \frac{\partial^2}{\partial y^2} f(\tau, y) \right) = 0 \tag{69}$$

> assume(tau1>0,x1>c);

Using the formula (48), we perform the following transformation of variables:

$$y' = \psi(\tau, y) = \ln(y - c) \Rightarrow$$

$$\Rightarrow \left[ \frac{1}{2} \frac{\partial^2}{\partial y^2} \psi(\tau, y) b(\tau, y) + \frac{\partial}{\partial y} \psi(\tau, y) a(\tau, y) + \frac{\partial}{\partial \tau} \psi(\tau, y) \right] = 0 \Rightarrow a'(\sigma', y') = 0,$$

$$\tau' = \varphi(\tau) = \tau^2 \Rightarrow \left\{ \left[ \frac{\partial}{\partial y} \psi(\tau, y) \right]^2 * b(\tau, y) \right\} : \frac{d}{d\tau} \varphi(\tau) = 1 \Rightarrow b'(\tau', y') = 1.$$

> tr:={y=c+exp(y1),tau=sqrt(tau1)};

$$tr := \{ \tau \sim = \sqrt{\tau 1 \sim}, y \sim = c \sim + e^{y1} \} \tag{70}$$

> PDE7CH:=dchange(tr,PDE7,[tau1,y1],simplify);

$$PDE7CH := \sqrt{\tau 1 \sim} \left( 2 \frac{\partial}{\partial \tau 1 \sim} f(\tau 1 \sim, y1 \sim, c) + \frac{\partial^2}{\partial y1^2} f(\tau 1 \sim, y1 \sim, c) \right) = 0 \tag{71}$$

> PDE7CHN:=expand (PDE7CH/(2\*sqrt(tau1)));

$$PDE7CHN := \frac{\partial}{\partial \tau 1 \sim} f(\tau 1 \sim, y1 \sim, c) + \frac{\frac{\partial^2}{\partial y1^2} f(\tau 1 \sim, y1 \sim, c)}{2} = 0 \tag{72}$$

> icch7:=f(0,y1,c)=Dirac(x1-y1);

$$icch7 := f(0, y1, c \sim) = Dirac(x1 \sim - y1) \tag{73}$$

The solution of the problem (72)-(73) is also expressed by the formula (63):

$$f(\tau 1, y1, x1, c) = \exp(-(y1 - x1)^2 / (2\tau 1)) / \text{sqrt}(2\pi\tau 1)$$

The resulting solution is the density of the normal distribution; we determine the density for the process under study:


$$f(\tau 1, y1, x1, c) [y1 = \psi(\tau, y), \tau 1 = \varphi(\tau)] * \frac{\partial}{\partial y} (\psi(\tau, y)) \Rightarrow f7(\tau, y, x1, c) = \frac{\exp\left(-\left(\tau^2 + x1 - \ln(y - c)\right)^2 / (2\tau^2)\right)}{\tau(y - c)\sqrt{2\pi}};$$

the last expression is not the density of the normal distribution ( this result is expected , since the transformation performed is not linear ) [Fig.12], [Fig.13].

> f7:=unapply(exp(-(tau^2+x1-ln(y-c))^2/(2\*tau^2))/(tau\*(y-c)\*sqrt(2\*Pi)),tau,y,x1,c);


$$f7 := (\tau, y, x1, c) \mapsto \frac{e^{-\frac{(\tau^2 + x1 - \ln(y - c))^2}{2\tau^2}} \sqrt{2}}{2\tau(y - c)\sqrt{\pi}} \tag{74}$$

> plot3d(subs({x1=0,c=1},f7(tau,y,x1,c)),tau=0.5..5,y=1..10,color=cos(tau),title="3D-SURFACE of f7 (tau,y,0,1)");



**Fig.12**

**> plot([subs({x1=0,c=1,tau=0.5},f7(tau,y,x1,c)),subs({x1=0,c=1,tau=1},f7(tau,y,x1,c))],y=1..10,thickness=5,color=[red,blue]);**




**Fig.13**

The animate graph below clearly shows the movement of the density surface depending on the parameter c [Fig. 14]:

**> animate(plot3d,[subs(x1=0,f7(tau,y,x1,c)),tau=0.5..1,y=c..10],c=1..5,style=patchcontour);**





**Fig.14**

#### **IV. CONCLUSIONS**

The article investigates the Fokker - Planck - Kolmogorov equation for a Markov random process. Various practically important cases are considered for the rate of change of the ordinate of the process and the rate of change of the conditional variance of the process. Analytical formulas for the densities of the corresponding distribution laws are obtained, 2D, 3D and animate graphs of densities are constructed. Using the capabilities of the Maple mathematical package it was possible to find connections between the known laws of probability distribution with various equations of the FPK-equation type. These studies mainly use the theory of Markov processes, the theory of probability distributions, the method of integral transformations, the method of characteristic functions, as well as the method of variable transformation.

#### **REFERENCES**

- [1]. Kolmogorov, A.A., (1938). "About analytical methods in probability theory, Journal of Successes of mathematical sciences, 5, 5-41 (in russian).
- [2]. Risken, H., (1984). "The Fokker - Planck equation", Springer, 452 p.
- [3]. Daniel, W.S., (2008). "Partial diff.equations for probability", Cambridge Studies in Advanced Mathematics, No 112.
- [4]. Sveshnikov, A.A., (2009). "Applied methods of the theory of random functions", Pergamon, 317 p.
- [5]. Bogachev, V.I., etc. (2015). "Fokker - Planck - Kolmogorov equations", Mathematical Surveys and Monographs, No 207, Rhode Island.
- [6]. [www.maplesoft.com/products/Maple2020/](http://www.maplesoft.com/products/Maple2020/)